
Processing
Beginners Workshop by Rosi Grillmair

Timetable
The program of this workshop would fit best in to a whole day workshop (8 – 10 hours) but can 
also be reduced to a 6 hour schedule. Then it would be a good idea to have time for Q&A and for 
the participants to try out their own ideas – within the presence of the instructor.

14:00 INTRO (1 h)
 Hello World
 Basics
 Text in Processing

//Break

15:00 Animation (2 h)
 Color
 Position

//Break

17:00 Interaction (2 h)
 Mouse
 Loops
 For Loop
 While Loop

//Break

19:00 Final Project: Lumis (1 h)
 What else can you do?

Intro

What is Processing?
Who developed it?
Who are the people behind it?
[core: Casey Reas, Ben Fry, a lot of contributors like Daniel Shiffman, Florian Jenett, Lauren 
McCarthy, platforms like Open Processing, taught in many schools]

• Open Processing
• hello processing http://hello.processing.org/

Some insights and projects that can be done with it:
• Princemio/onformative: Pathfinder http://princemio.net/portfolio/pathfinder/ 
• Espadaysantacruz: Kinetic Light Sculpture http://espadaysantacruz.com/Light-kinetics 
• LIA: Filament Sculptures: http://www.liaworks.com/theprojects/filament-sculptures/
• Dextro: Non Linear Code http://dextro.org/default_h.html 

Requirements
Operating System: any is okay (Windows, Linux, OS X)
Download the latest stable release from processing.org/download/?processing

Resources
• Code Editor
• Reference
• Possible Modes: Javascript, p5, Python, Android, Arduino

What will we be able to do in the end?
The Lumi Project (see last piece of code)

Let's start sketching!

1# hello world – print it
→ print it to the text area

What is a function? How to call a function? What is its input? (=parameter), command println()

println("hello world");

#2 basics -   Display Window and how to work with it

→ make window 300 x 300 and draw quad with size 100 on a red background
size(), rect(), background(), coordinate system, measurement in pixels, color selector, RGB and Hex 
values for colors, always end a command with ;

size(300, 300);
background(255, 0, 0);
rect(0, 0, 100, 100);

→ put quad to the center
use rectMode(), width and height

size(300, 300);
background(255, 0, 0);
rectMode(CENTER);
rect(width/2, height/2, 100, 100);

→ change color of quad
use fill()

→ use variables for the position of the quad
xPos, yPos, declaration, assignments and usecase
data types for numbers: int, float, long, double
Order of the commands is important! What happens if you put the variable assignment to the 
beginning?

size(300, 300);
int xPos = width/2;
int yPos = height/2;
background(255, 0, 0);
rectMode(CENTER);
rect(xPos, yPos, 100, 100);

→ make a second quad and play around with commands and values, rearrange the commands, 
make a comments before you draw the quad to the screen: “quad 1”, “quad 2”
how to write comments

#3 hello world – print to Display Window
→ try to print “hello world” to the screen on a display window size 100x100
tips: use the function text() and textAlign(), don't forget quotation marks, otherwise the program 
thinks it's a variable and not a text

//Solution
size(100, 100);
textAlign(LEFT, TOP);
text("hello world", 0, 0);

#4 Text in Processing
→ declare a variable “s1” and assign “hello world” and then draw the text to the screen, also use 
variables for positions again
data types for text: String, char

size(100, 100);
int xPos = 0;
int yPos = 0;
String s1 = "hello world";
textAlign(LEFT, TOP);
text(s1, xPos, yPos);

→ have a look on the documentation and decide on a thing you want to try. Font? Fill? Get 
letters from text?
https://processing.org/reference/text_.html

//BREAK

Animation
#5 Animating a Color Value
→ draw a circle to the center of the screen without stroke line and make it change the color from 
white to black
+ “smooth” the outlines.
+ print the value of the variable that changes the color to the screen
+ change the speed to 10 and then 100 fps
setup() and draw(), greyscale color values, ellipse() and ellipseMODE(), noStroke(), stroke(), 
smooth(), frameRate(), increase a value using ++, what is camelCase?

int greyValue;

void setup() {
  size(300, 300);
  ellipseMode(CENTER);
  noStroke();
  smooth();
  frameRate(100);
  greyValue = 255;
}

void draw() {
  println(greyValue);
  fill(greyValue);
  ellipse(width/2, height/2, 100, 100);
  greyValue--;
}

→ now let the greyValue start at 255 and let it go from white to black
decreasing a value using –

int greyValue;

void setup() {
  //...

  greyValue = 255;
}

void draw() {
  //...
  greyValue--;
}

→ but what when greyValue is beyond 255? Make it jump back to 0.
+ make it twice as fast but not by changing the fps, but by making the steps bigger in which the 
greyValue changes
Conditions: if statement, add and subtract values using += and -=

//...

void draw() {
  println(greyValue);
  fill(greyValue);
  ellipse(width/2, height/2, 100, 100);
  greyValue+=2;
  if(greyValue > 255){
    greyValue = 0;
  }
}

#6 Animating the Position of an Object

→ try to make the circle move from left to right (but stay in the middle of the screen). When it 
leaves the screen on the right it should appear again on the left. Which variable needs to be 
changed?
background() function within draw(), assigning more than one variable of the same data type
int xPos, yPos;

void setup() {
  size(500, 500);
  noStroke();
  smooth();
  frameRate(100);
  
  xPos = 0;
  yPos = height/2;  
}

void draw() {
  background(0);
  ellipse(xPos, yPos, 100, 100);
  xPos++;
  if(xPos > width){
    xPos = 0;
  }
}

→ now let the ball bounce between left and right! We need math for that.
 use OR (= ||) in a conditional statement

int xPos, yPos;
int dir;

void setup() {
  size(500, 500);
  noStroke();
  smooth();
  frameRate(100);
  
  xPos = 0;
  yPos = height/2;
  dir = 1;  
}

void draw() {
  background(0);
  ellipse(xPos, yPos, 100, 100);
  xPos+= 1*dir;
  if(xPos <0 || xPos > width){
    dir*= -1;
  }
}

→ we want it to bounce from its side not from the center. How could we do that? (Pay attention to 
where the start value of the x Position is!)

int xPos, yPos;
int dir;
int rad;

void setup() {
  size(500, 500);
  noStroke();
  smooth();

  xPos = width/2;
  yPos = height/2;
  dir = 1;  
  rad = 50;
}

void draw() {
  background(0);
  ellipse(xPos, yPos, 50*2, 50*2);
  xPos+= 2*dir;
  if (xPos <rad || xPos > width-rad) {
    dir*= -1;
  }
}

→ now try to make it bounce also on the y – axis! It should bounce from the upper left corner to 
the lower right corner.

int xPos, yPos;
int dirX, dirY;
int rad;

void setup() {
  size(500, 500);
  noStroke();
  smooth();
  xPos = width/2;
  yPos = height/2;
  dirX = 1; 
  dirY = 1; 
  rad = 50;
}

void draw() {
  background(0);
  ellipse(xPos, yPos, 50*2, 50*2);
  xPos+= 1*dirX;
  yPos+= 1*dirY;
  if (xPos <rad || xPos > width-rad) {
    dirX*= -1;
  }
  if (yPos <rad || yPos > width-rad) {
    dirY*= -1;
  }
}

→ It's a bit boring! We will let it bounce randomly. What does it do when you just put the random 
function into draw()? How should you use it to make it bounce?

The random() Function: It chooses a random value in between a range that is defined by you. 
random(5) will find a value between 0 and 5, random (2, 5) will find a value between 2 and 5.

float xPos, yPos;
int dirX, dirY;
float stepX, stepY;
int rad;

void setup() {
  size(500, 500);
  noStroke();
  smooth();

  xPos = width/2;
  yPos = height/2;
  dirX = 1; 
  dirY = 1;
  stepX =  random(-5, 6);
  stepY =  random(-5, 6);
  rad = 50;
}

void draw() {
  background(0);
  ellipse(xPos, yPos, 50*2, 50*2);
  xPos+= stepX*dirX;
  yPos+= stepY*dirY;
  if (xPos <rad || xPos > width-rad) {
    dirX*= -1;
  }
  if (yPos <rad || yPos > width-rad) {
    dirY*= -1;
  }
}

→ change the color to a random value everytime the ball bounces!

//...
if (xPos <rad || xPos > width-rad) {
    dirX*= -1;
    fill(random(255), random(255), random (255));
  }
  if (yPos <rad || yPos > width-rad) {
    dirY*= -1;
     fill(random(255), random(255), random (255));
  }
//...

→ Would the whole thing work with a quad as well? What are the 2 changes you have to make?

void setup(){
 //...
 rectMode(CENTER);
}

void draw() {
 //...
 rect(xPos, yPos, 50*2, 50*2);
 //...
}

//BREAK

#7 Interactions: Mouse Position
→ make a “drawing program” by attaching a circle or square to the mouse position. It looks nicer if 
you make the cursor invisible and put away the stroke line of the object.
Variables mouseX and mouseY, noCursor()

void setup(){
  size(800, 800);
  noCursor();
  noStroke();
}

void draw(){
  ellipse(mouseX, mouseY, 50, 50);
}

→ choose a color and change the transparency value (Alpha channel)
transparency values and the Alpha channel

void setup(){
  //...
  fill(255, 200, 10, 100);
}

→ now make the object a cursor
void draw(){
  background(255);
  //..
}

→ now we are going to change the color of the background by moving the mouse around.
Mapping function: map() a value to another range

void setup(){
  size(800, 800);
}

void draw(){
  background(map(mouseX, 0, width, 0, 255), map(mouseY, 0, height, 0, 
255), 0);
}

→ Reuse the code of your “drawing program” and change the color of the pencil according to its 
position on the screen.

void setup(){
  size(800, 800);
  noCursor();
  noStroke();
}

void draw(){
  fill(map(mouseX, 0, width, 0, 255), map(mouseY, 0, height, 0, 255), 0);
  ellipse(mouseX, mouseY, 50, 50);
}

→ change the color of a circle that is placed in the middle of the screen by mouse position: make it
white when the cursor is in the middle of it and black if it's as far away as possible. - some math 
involved again! Print the values to the text areas to make sure.
use dist() and map() method

int xPos, yPos;

void setup(){
  size(800, 800);
  noStroke();
  xPos = width/2;
  yPos = height/2;
}

void draw(){
  background(0);
  println(dist(xPos, yPos, mouseX, mouseY));
  println(map(dist(xPos, yPos, mouseX, mouseY), 0, dist(xPos, yPos, 
width, height), 255, 0));
  fill(map(dist(xPos, yPos, mouseX, mouseY), 0, dist(xPos, yPos, width, 
height), 255, 0));
  ellipse(xPos, yPos, 100, 100);
}

→ now also change the size so the ellipse reaches the side of the display window when the cursor 
is in the center! Use a variable type int called “diameter”. What error occurs when you try to run 
it?
Casting data types: sometimes it's necessary to change a data type, like a float to an int or vice 
versa. To do so write the data type in front of the value you need to change to another type: (int)...
or (float)...

//Example
float x = 6.5;
int y = (int)x;
println(y);
//prints: 6

map always gives a float, to assign the map-value to a variable of the type int, you have to cast it.
or: make "rad" a variable of the data type float from the beginning

int xPos, yPos;
int diameter;

void setup(){
  size(800, 800);
  noStroke();
  xPos = width/2;
  yPos = height/2;
  diameter = 100;
}

void draw(){
  background(0);
  fill(map(dist(xPos, yPos, mouseX, mouseY), 0, dist(xPos, yPos, width, 
height), 255, 0));  
  diameter = (int)map(dist(xPos, yPos, mouseX, mouseY), 0, dist(xPos, 
yPos, width, height), width, 0);
  ellipse(xPos, yPos, diameter, diameter);
}

11



→ Now we want the circle to be white only when the cursor is above it, otherwise it should be 
black. Remember to consider the radius and work with a variable “rad”.
Conditional statement:  use AND (=&&), make an if and an else condition

int xPos, yPos;
int rad;
void setup(){
  size(800, 800);
  noStroke();
  xPos = width/2;
  yPos = height/2;
  rad = 50;
}
void draw(){
  background(0);
  if(mouseX > xPos-rad && mouseX<xPos+rad && mouseY > yPos-rad && 
mouseY<yPos+rad){
    fill(255);
  }else{
    fill(0);
  }
  ellipse(xPos, yPos, rad*2, rad*2);
}

→ try to divide the display window into 2 areas, make it green when the mouse is above it and 
grey if it's not. 

//Solution 1 (slow)
void setup() {
  size(800, 800);
}

void draw() {
  if (mouseX<width/2) {
    fill(0, 255, 0);
    rect(0, 0, width/2, height);
    fill(125);
    rect(width/2, 0, width/2, height);
  }
  else {
    fill(125);
    rect(0, 0, width/2, height);
    fill(0, 255, 0);
    rect(width/2, 0, width/2, height);
  }
}

//Solution 2 (qick)
void setup(){
  size(800, 800);
}

void draw(){
  background(125);
  fill(0, 255, 0);
  if(mouseX<width/2){
    rect(0, 0, width/2, height);
  }
  else{
  rect(width/2, 0, width/2, height);
  }
}

→ make the stroke line of the rectangles invisible, instead make a line which is thick 2p and white
use the line Function: line(x1, y1, x2, y2), strokeWeight() and stroke(color)
void setup(){
  size(800, 800);
}

void draw(){
  background(125);
  noStroke();
  fill(0, 255, 0);
  if(mouseX<width/2){
    rect(0, 0, width/2, height);
  }
  else{
  rect(width/2, 0, width/2, height);
  }
  stroke(255);
  strokeWeight(2);
  line(width/2, 0, width/2, height);
}

#8 many things at the same time: Loops

→ create 20 circles with random colors and random diameters anywhere on the screen
for – loop: define how often you want to repeat something

void setup(){
  size(800, 800);
  for(int i =0; i<10; i++){
    fill(random(255), random(255), random(255));
    float diameter = random(20, 100);
    ellipse( random(width), random(height), diameter, diameter);
  }
}

→ try the same thing again but use a while – loop!
while – loop: repeat something until a condition changes

void setup(){
  size(800, 800);
  int i = 0;
  while(i<10){
    fill(random(255), random(255), random(255));
    float diameter = random(20, 100);
    ellipse( random(width), random(height), diameter, diameter);
    i++;
  }
}

→ create a display window that's 800 x 800. In one row (10 px from the top) place 20 squares with 
a distance of 10 px to each other. How big are the squares when the first and the last touch the 
side of the screen?
Use the variables amount (int), squareSize (float), dist (float)!

How do you reach the equation squareSize = (width - amount*dist + dist)/amount?
width = amount* squareSize + (amount – 1)*dist
amount* squareSize = width – amount*dist + dist
squareSize = (width - amount*dist + dist)/amount

int amount;
float dist, squareSize;

void setup() {
size(800, 800);

amount = 20;
  dist = 10;
  squareSize = (width - amount * dist + dist)/amount;
  for (int i = 0; i<amount; i++) {
      rect(i*(squareSize+dist), dist, squareSize, squareSize);
  }
}

→ try the same again but use a while Loop

→ now fill the screen with squares!
Put 2 for – loops within each other.

int amount;
float dist, squareSize;

void setup() {
  size(800, 800);

  amount = 20;
  dist = 10;
  squareSize = (width - amount * dist + dist)/amount;
  for (int i = 0; i<amount; i++) {
    for (int j = 0; j<amount; j++) {
      rect(i*(squareSize+dist), j*(squareSize+dist),

 squareSize, squareSize);
    }
  }
}

→ let the size change according to mouseX movement (from 0 to width/amount )

int amount;
float dist, squareSize;

void setup() {
  size(800, 800);
  
}

void draw(){
  background(0);
  amount = 20;
  dist = map(mouseX, 0, height, 0, width/amount);
  squareSize = (width - amount * dist + dist)/amount;
  for (int i = 0; i<amount; i++) {
    for (int j = 0; j<amount; j++) {
      rect(i*(squareSize+dist), j*(squareSize+dist), squareSize, 
squareSize);
    }
  }
}

→ now let the size get bigger when its near the mouse!
int amount;
float squareSize, dist;

void setup() {
  size(800, 800);
  noStroke();
}

void draw() {
  background(0);
  amount = 20;
  dist = width/amount;
  for (int i = 0; i<amount; i++) {
    for (int j = 0; j<amount; j++) {      
      squareSize = map(dist(i*dist, j*dist, mouseX, mouseY), 0, dist(0, 

 0, width, height), dist, 0);
      rect(i*dist+squareSize/2, j*dist+squareSize/2, squareSize,  

 squareSize);
    }
  }
}

→ try the following: make circles that all change their color when the mouse is near!

int amount;
float size;

void setup() {
  size(800, 800);
  noStroke();
}

void draw() {
  background(0);
  amount = 20;
  size = width/amount;
  for (int i = 0; i<amount; i++) {
    for (int j = 0; j<amount; j++) {      
      fill(map(dist(i*size, j*size, mouseX, mouseY), 0, dist(0, 0, width,
height), 255, 0));
      ellipse(i*size+size/2, j*size+size/2, size, size);
    }
  }
}

→ how could you change the radius of the visible ellipses efficiently? (only 1 number)

//...
fill(map(dist(i*size, j*size, mouseX, mouseY), 0, dist(0, 0, width, 
height)/5, 255, 0));
//...

→ create as many lines next to each other as possible, each of them having another color value to 
represent the whole sprectrum using HSB color mode 

HSB: Hue, Saturation, Brightness – also range between 0, 255 but it's possible to change the range 
using colorMode – z.B. colorMode(HSB, 500);
void setup() {
  size(800, 800);
  colorMode(HSB, width);
  
  for (int i = 0; i<width; i++) {
    stroke(i, width, width);
    line(i, 0, i, height);
  }
}

→ also include the saturation value and use a point instead of a line to make every point have 
another color value.

void setup() {
  size(255, 255);
  colorMode(HSB, width);

  for (int i = 0; i<width; i++) {
    for (int j = 0; j<height; j++) {
      stroke(i, j, width);
      point(i, j);
    }
  }
}

#9 Project: Lumis

→ a swarm of little light balls are exploring the dark. Maybe they find a hidden thing.. a source of 
energy maybe? Help the swarm collect them!
Combination of random objects, motion, events according to mouse movements and many objects
at one.
New way to create a background (with alpha value): full a rect that is the same size as the screen!
How to make a blurry light balls? Overlay ellipses of different size!

//hidden thing
float hX, hY;
int visible, findRange;

// lumis
float range, posX, posY;
int size, amount;

void setup() {
  size(800, 800);
  background(0);
  noStroke();
  frameRate(30);
  noCursor();

  //hidden thing
  hX = random(width);
  hY = random(height);
  visible = 10;  
  findRange = 10;

  //lumis
  size = 25;
  amount = 10;
}

void draw() {
  //background
  fill(0, 40);
  rect(0, 0, width, height);

  //lumis  
  for (int i=0; i<amount; i++) { 
    float hiddenDist = dist(mouseX, mouseY, hX, hY);
    fill(255, 255, map(hiddenDist, 0, 
    dist(0, 0, width/visible, height/visible), 0, 255), 255/size);  
    range = random(10, 50);
    posX = random(mouseX-range, mouseX+range);
    posY  =random(mouseY-range, mouseY+range);
    for (int j=0; j<size; j++) {
      ellipse(posX, posY, size-j, size-j);
    }
  }
  if (mouseX >= hX-findRange && mouseX <= hX+findRange && mouseY >= hY-
findRange && mouseY <= hY+findRange) {
    hX = random(width);
    hY = random(height);
  }
}

→ create one little light ball that follows the mouse has a blur to it.

// lumis
float range, posX, posY;
int size, amount;

void setup() {
  size(800, 800);
  background(0);
  noStroke();
  frameRate(30);
  noCursor();
  //lumis
  size = 25;
  amount = 10;
}

void draw() {
  background(0);  
  //lumis  
  posX = mouseX;
  posY = mouseY;
  for (int j=0; j<size; j++) {
    fill(255, 255/size);
    ellipse(posX, posY, size-j, size-j);
  }
}

→ add a motion blur effect: replace the background with a slightly transparent fullscreen quad.

  //...
  //background
  fill(0, 40);
  rect(0, 0, width, height);
  //..

→ make a swarm of 10 light balls (variable: amount) and let them appear in a range which can be 
between 10 and 50 around the mouse position!

//...
//lumis  
  for (int i=0; i<amount; i++) {   
    range = random(10, 50);
    posX = random(mouseX-range, mouseX+range);
    posY  =random(mouseY-range, mouseY+range);
    for (int j=0; j<size; j++) {
      fill(255, 255/size);
      ellipse(posX, posY, size-j, size-j);
    }
  }
//...

→ now we only have to add the hidden thing. Assign a random position to it, then let the light balls
change color when they come near it!

Add a view variables on the top:
//hidden thing
float hX, hY;
int visible, findRange;
//...

initalize them in the setup function:

  //hidden thing
  hX = random(width);
  hY = random(height);
  visible = 10;  
  findRange = 10;

and then edit the fill function as we already did before (don't forget to delete the other fill function
within the second loop!)

  for (int i=0; i<amount; i++) { 
    float hiddenDist = dist(mouseX, mouseY, hX, hY);
    fill(255, 255, map(hiddenDist, 0, 
    dist(0, 0, width/visible, height/visible), 0, 255), 255/size);  
    range = random(10, 50);
    posX = random(mouseX-range, mouseX+range);

    posY  =random(mouseY-range, mouseY+range);
    for (int j=0; j<size; j++) {
      ellipse(posX, posY, size-j, size-j);
    }
  }

→ now add a condition that when you reach the  “findRange” - here it is 10 – with the swarm, the 
position of the hidden thing should become a new one.

if (mouseX >= hX-findRange && mouseX <= hX+findRange &&
mouseY >= hY-findRange && mouseY <= hY+findRange) {
    hX = random(width);
    hY = random(height);
  }
//end of draw function
}

More Programming Concepts:
• Classes and Objects
• Arrays and Lists
• Transformations: Translate, Scale, Rotate
• P3D: 3D Space, Forms, Transformations, Textures, Lighting, Shaders
• Pgraphics, Pshapes
• Vectors, Pvector Library, Flow Fields
• UDP Communication
• Particle Systems

Interfaces:
• Video Effects
• Sound Library (Minim)
• Android Apps
• Arduino Firmata Library
• Realtime Online Data
• Color Tracking
• Fiducial Tracking
• OpenCV – Standard Tracking

Resources:
The Nature of Code by Daniel Shiffman -  https://vimeo.com/channels/natureofcode, 
https://vimeo.com/83399009 (Laurin)
p5 - http://p5js.org/ 
Reference: https://processing.org/reference/ 
Java Reference: http://docs.oracle.com/javase/7/docs/api/overview-summary.html 

